The Gross-Pitaevskii equation and Bogoliubov spectrum

Pascal Szriftgiser[1] and Edgardo S. Cheb-Terrab[2]
(1) Laboratoire PhLAM, UMR CNRS 8523, Université Lille 1, F-59655, France
(2) Maplesoft, Canada

Departing from the equation for a quantum system of identical boson particles, i.e.the Gross-Pitaevskii equation (GPE), the dispersion relation for plane-wave solutions are derived, as well as the Bogoliubov equations and dispersion relations for small perturbations $\delta\varphi$ around the GPE stationary solutions.

Stationary and plane-wave solutions to the Gross-Pitaevskii equation

Problem: Given the Gross-Pitaevskii equation,

$$i \hbar \psi_t = \left(G |\psi|^2 + V\right) \psi - \frac{\hbar^2}{2 m} \nabla^2\psi$$

a) Derive a relationship between the chemical potential μ entering the phase of stationary, uniform solutions, the atom-atom interaction constant G and the particle density $n = |\psi|^2$ in the absence of an external field ($V = 0$).

b) Derive the dispersion relation for plane-wave solutions as a function of G and n.

Background: The Gross-Pitaevskii equation is particularly useful to describe Bose Einstein condensates (BEC) of cold atomic gases [3, 4, 5]. The temperature of these cold atomic gases is typically in the ~100 nano-Kelvin range. The atom-atom interaction are repulsive for $G > 0$ and attractive for $G < 0$, where G is the interaction constant. The GPE is also widely used in non-linear optics to model the propagation of light in optical fibers.

Solution

a) Derive a relationship between the chemical potential μ entering the phase of stationary, uniform solutions, the atom-atom interaction constant G and the particle density $n = |\psi|^2$ in the absence of an external field ($V = 0$).

The version of Physics used is from December/6 (or later), available at the Maplesoft Physics Research & Development webpage

```
> restart; with(Physics) : with(Physics[Vectors]) :
> interface(imaginaryunit = i) :
> Setup(mathematicalnotation = true)
```

$$[mathematicalnotation = true] \qquad \textbf{(1.1.1)}$$

Use a macro $\Psi = \psi(x, y, z, t)$ to avoid redundant typing and use declare to have a compact display

```
> macro(Psi = psi(x, y, z, t)) :
```

> *PDEtools*:-*declare*((ψ, V) (x, y, z, t))

$\psi(x, y, z, t)$ *will now be displayed as* ψ

$V(x, y, z, t)$ *will now be displayed as* V **(1.1.2)**

The Gross-Pitaevskii equation, derived in the previous post in Mapleprimes, is given by

> $i\,\hbar\, diff(\mathrm{Psi}, t) = \left(G\left(\mathrm{abs}(\mathrm{Psi})^2\right) + V(x, y, z, t)\right)\mathrm{Psi} - \hbar^2\ \%Laplacian(\mathrm{Psi}) / (2\,m);$

$$\mathrm{i}\,\hbar\,\psi_t = \left(G\,|\psi|^2 + V\right)\psi - \frac{\hbar^2\,\nabla^2\psi}{2\,m} \quad \textbf{(1.1.3)}$$

where G is the atom-atom interaction constant, $V(x, y, z, t)$ is an external field, ψ is the particle-wave function, m is the mass of a single particle and $\hbar$ is the Planck constant divided by 2π. Set the real objects of this problem including the chemical potential μ

> *Setup*(*realobjects* = { $m, t, G, V(x, y, z, t), \hbar$, mu })

$$\left[realobjects = \left\{\hbar, G, \hat{i}, \hat{j}, \hat{k}, \hat{\phi}, \hat{r}, \hat{\rho}, \hat{\theta}, m, \mu, \phi, r, \rho, t, \theta, x, y, z, V\right\}\right] \quad \textbf{(1.1.4)}$$

In the absence of an external potential, we have $V(x, y, z, t) = 0$. To relate the chemical potential to the interaction constant G and the particle density n we look for a real stationary and uniform solution for **(1.1.3)** of the form $\psi = \varphi\, \mathrm{e}^{-\frac{\mathrm{i}\cdot\mu\, t}{\hbar}}$, where μ is the chemical potential and φ is constant.

> $\psi(x, y, z, t) = \varphi\, \mathrm{e}^{-\frac{\mathrm{i}\,\mu\, t}{\hbar}}$

$$\psi = \varphi\, \mathrm{e}^{\frac{-\mathrm{i}\,\mu\, t}{\hbar}} \quad \textbf{(1.1.5)}$$

Substituting into the GPE equation **(1.1.3)**,

> *eval*(**(1.1.3)**, { **(1.1.5)**, $V(x, y, z, t) = 0$ })

$$\varphi\,\mu\,\mathrm{e}^{\frac{-\mathrm{i}\,\mu\, t}{\hbar}} = G\,|\varphi|^2\,\varphi\,\mathrm{e}^{\frac{-\mathrm{i}\,\mu\, t}{\hbar}} - \frac{\hbar^2\,\nabla^2\left(\varphi\,\mathrm{e}^{\frac{-\mathrm{i}\,\mu\, t}{\hbar}}\right)}{2\,m} \quad \textbf{(1.1.6)}$$

The relationship between μ, G and n is then obtained evaluating the Laplacian in the right-hand side, isolating μ and introducing the particle density $n = |\psi|^2$

> *isolate*(*value*(**(1.1.6)**), mu)

$$\mu = G\,|\varphi|^2 \quad \textbf{(1.1.7)}$$

From **(1.1.5)**,

> *map*($u \rightarrow \mathrm{abs}(u)^2$, **(1.1.5)**)

$$|\psi|^2 = |\varphi|^2 \quad \textbf{(1.1.8)}$$

> *subs*($\mathrm{abs}(\mathrm{Psi})^2 = n$, **(1.1.8)**)

$$n = |\varphi|^2 \quad \textbf{(1.1.9)}$$

From where

> *subs*((*rhs* = *lhs*) (**(1.1.9)**), **(1.1.7)**)

$$\mu = G\,n \quad \textbf{(1.1.10)}$$

b) Derive the dispersion relation for plane-wave solutions as a function of G and n.

We now look for plane-wave solutions $\sqrt{n}\, e^{i k x - I \omega t}$ where k is the wave number and ω the frequency. To simplify, the GPE is restricted to a 2D spacetime case, still in the absence of external potential.

```
> macro(Psi2 = psi(x, t)) :
> PDEtools:-declare(Psi2)
```

$\psi(x, t)$ *will now be displayed as* ψ **(1.1.11)**

```
> Setup(realobjects = {n, k, omega})
```

$$[realobjects = \{\hbar, G, \hat{i}, \hat{j}, \hat{k}, \hat{\phi}, \hat{r}, \hat{\rho}, \hat{\theta}, k, m, \mu, n, \omega, \phi, r, \rho, t, \theta, x, y, z, V\}] \quad (1.1.12)$$

So the 2D GPE at $V = 0$ is obtained replacing $\psi(x, y, z, t) = \psi(x, t)$ in **(1.1.3)**

```
> subs(V(x, y, z, t) = 0, psi(x, y, z, t) = psi(x, t), (1.1.3))
```

$$i \hbar \psi_t = G \psi |\psi|^2 - \frac{\hbar^2 \nabla^2 \psi}{2 m} \quad (1.1.13)$$

Introducing now an explicit form of ψ in terms of the particle density $n = |\psi|^2$ and a plane wave $e^{i k x - I \omega t}$, and taking into account that $n \geq 0$,

```
> Psi2 = sqrt(n) exp(i (k x - omega t))
```

$$\psi = \sqrt{n}\, e^{i (k x - \omega t)} \quad (1.1.14)$$

```
> eval((1.1.13), (1.1.14)) assuming n ≥ 0
```

$$\hbar \sqrt{n}\, \omega e^{i (k x - \omega t)} = G n^{3/2} e^{i (k x - \omega t)} - \frac{\hbar^2 \nabla^2\left(\sqrt{n}\, e^{i (k x - \omega t)}\right)}{2 m} \quad (1.1.15)$$

Isolating ω and evaluating the Laplacian we get the dispersion relation

```
> isolate((1.1.15), omega)
```

$$\omega = \frac{2 G n e^{i k x} m - \hbar^2 \nabla^2 e^{i k x}}{2 \hbar e^{i k x} m} \quad (1.1.16)$$

```
> expand(value(%))
```

$$\omega = \frac{G n}{\hbar} + \frac{\hbar k^2}{2 m} \quad (1.1.17)$$

This GPE dispersion relation is a function of both the non-linearity G and the field density n (which would not be the case for a linear PDE).

The Bogoliubov equations and dispersion relations

Problem: Given the Gross-Pitaevskii equation,

a) Derive the Bogoliubov equations, that is, equations for elementary excitations $\delta\varphi$ and $\overline{\delta\varphi}$ around a GPE stationary solution $\varphi(x, y, z)$,

$$\begin{cases} i\hbar \dfrac{\partial}{\partial t}\,\delta\varphi = -\dfrac{\hbar^2\,\nabla^2\delta\varphi}{2\,m} + \left(2\,G\,|\varphi|^2 + V - \mu\right)\delta\varphi + G\,\varphi^2\,\overline{\delta\varphi} \ , \\ i\hbar \dfrac{\partial}{\partial t}\,\overline{\delta\varphi} = +\dfrac{\hbar^2\,\nabla^2\overline{\delta\varphi}}{2\,m} - \left(2\,G\,|\varphi|^2 + V - \mu\right)\overline{\delta\varphi} - G\,\delta\varphi\,\overline{\varphi}^2 \ , \end{cases}$$

b) Show that the dispersion relations of these equations, known as the Bogoliubov spectrum, are given by

$$\varepsilon_k = \hbar\,\omega_k = \pm\sqrt{\frac{\hbar^4\,k^4}{4\,m^2} + \frac{\hbar^2\,k^2\,G\,n}{m}}\ ,$$

where k is the wave number of the considered elementary excitation, ε_k its energy or, equivalently, ω_k its frequency.

Solution

a) To derive Bogoliubov's equations, start from a non-uniform stationary GPE solution, so $|\psi|$ depending only on x, y, z, that is, of the form $\psi = \varphi(x, y, z)\,e^{-\frac{i\,\mu\,t}{\hbar}}$, where written in this form μ is the chemical potential, and introduce small perturbations $\delta\varphi(x, y, z, t)$ around $\varphi(x, y, z)$. Bogoliubov's equations describe the evolution in time and space of these small perturbations $\delta\varphi$ and are obtained, basically, by inserting the perturbed solution into the GPE and discarding terms of higher degree in $\delta\varphi$.

> $PDEtools\text{:-}declare\big(\psi(x, y, z, t), \varphi(x, y, z), \delta\varphi(x, y, z, t)\big)$

$\psi(x, y, z, t)$ *will now be displayed as* ψ

$\varphi(x, y, z)$ *will now be displayed as* φ

$\delta\varphi(x, y, z, t)$ *will now be displayed as* $\delta\varphi$ **(2.1.1)**

The departing point is then the form of the GPE solution $\psi(x, y, z, t)$ that includes this small perturbation $\delta\varphi(x, y, z, t)$ depending on space and time

> $\mathrm{Psi} = \big(\varphi(x, y, z) + \delta\varphi(x, y, z, t)\big)\exp\left(-\frac{i\,\mathrm{mu}\,t}{\hbar}\right)$

$$\psi = (\varphi + \delta\varphi)\,e^{\frac{-i\,\mu\,t}{\hbar}} \tag{2.1.2}$$

Recalling the form of the GPE **(1.1.3)**

> **(1.1.3)**

$$i\,\hbar\,\psi_t = \left(G\,|\psi|^2 + V\right)\psi - \frac{\hbar^2\,\nabla^2\psi}{2\,m} \tag{2.1.3}$$

Substitute into it the perturbed ψ

> $eval(\textbf{(1.1.3)}, \textbf{(2.1.2)})$

$$i\hbar\left(\delta\varphi_t\, e^{\frac{-i\mu t}{\hbar}} - \frac{i(\varphi+\delta\varphi)\,\mu\, e^{\frac{-i\mu t}{\hbar}}}{\hbar}\right) = \left(G|\varphi+\delta\varphi|^2 + V\right)(\varphi+\delta\varphi)\, e^{\frac{-i\mu t}{\hbar}} - \frac{\hbar^2\nabla^2\left((\varphi+\delta\varphi)\, e^{\frac{-i\mu t}{\hbar}}\right)}{2m} \quad \textbf{(2.1.4)}$$

> $expand\left(\textbf{(2.1.4)}\ \exp\left(\frac{i\,\mathrm{mu}\,t}{\hbar}\right)\right)$

$$i\hbar\,\delta\varphi_t + \mu\varphi + \mu\,\delta\varphi = G|\varphi+\delta\varphi|^2\varphi + G|\varphi+\delta\varphi|^2\delta\varphi + V\varphi + V\delta\varphi - \frac{\hbar^2\nabla^2\varphi}{2m} - \frac{\hbar^2\nabla^2\delta\varphi}{2m} \quad \textbf{(2.1.5)}$$

The unperturbed GPE solution $\varphi(x, y, z)\cdot e^{-\frac{i\mu t}{\hbar}}$ is also substituted into **(2.1.3)**:

> $\psi(x, y, z, t) = \varphi(x, y, z)\ \exp\left(-\frac{i\,\mathrm{mu}\,t}{\hbar}\right)$

$$\psi = e^{\frac{-i\mu t}{\hbar}}\varphi \quad \textbf{(2.1.6)}$$

> $eval(\textbf{(2.1.3)}, \textbf{(2.1.6)})$

$$\mu\, e^{\frac{-i\mu t}{\hbar}}\varphi = \left(G|\varphi|^2 + V\right) e^{\frac{-i\mu t}{\hbar}}\varphi - \frac{\hbar^2\nabla^2\left(e^{\frac{-i\mu t}{\hbar}}\varphi\right)}{2m} \quad \textbf{(2.1.7)}$$

> $expand\left(\textbf{(2.1.7)}\ e^{\frac{i\mu t}{\hbar}}\right)$

$$\mu\varphi = \varphi G|\varphi|^2 + V\varphi - \frac{\hbar^2\nabla^2\varphi}{2m} \quad \textbf{(2.1.8)}$$

To obtain the equation for the perturbation $\delta\varphi$, subtract the non-perturbed solution **(2.1.8)** from the perturbed solution **(2.1.5)**:

> **(2.1.5)** - **(2.1.8)**

$$i\hbar\,\delta\varphi_t + \mu\,\delta\varphi = G|\varphi+\delta\varphi|^2\varphi + G|\varphi+\delta\varphi|^2\delta\varphi + V\delta\varphi - \frac{\hbar^2\nabla^2\delta\varphi}{2m} - \varphi G|\varphi|^2 \quad \textbf{(2.1.9)}$$

In order to discard higher degree terms in the perturbation $\delta\varphi$ and $\overline{\delta\varphi}$, rewrite the $|\varphi+\delta\varphi|^2$ in terms of $\varphi+\delta\varphi$ and its conjugate

> $expand(convert(\textbf{(2.1.9)}, conjugate))$

$$i\hbar\,\delta\varphi_t + \mu\,\delta\varphi = \varphi^2 G\,\overline{\delta\varphi} + 2\varphi G\,\delta\varphi\,\overline{\varphi} + 2\varphi G\,\delta\varphi\,\overline{\delta\varphi} + G\,\delta\varphi^2\,\overline{\varphi} + G\,\delta\varphi^2\,\overline{\delta\varphi} + V\delta\varphi - \frac{\hbar^2\nabla^2\delta\varphi}{2m} \quad \textbf{(2.1.10)}$$

> $simplify(\textbf{(2.1.10)}, size)$

$$i\hbar\,\delta\varphi_t + \mu\,\delta\varphi \quad \textbf{(2.1.11)}$$

$$= \frac{2\, G\, m\, (\varphi + \delta\varphi)^2\, \overline{\delta\varphi} - \hbar^2\, \nabla^2 \delta\varphi + 2\, \delta\varphi\, m\, \left(G\, \overline{\varphi}\, \delta\varphi + 2\, G\, \overline{\varphi}\, \varphi + V\right)}{2\, m}$$

In **(2.1.11)** we see that, discarding higher degree terms in $\delta\varphi$ and $\overline{\delta\varphi}$, we have $(\varphi + \delta\varphi)^2\, \overline{\delta\varphi} \approx \varphi^2\, \overline{\delta\varphi}$, and $\delta\varphi\, G\, \overline{\varphi}\, \delta\varphi \approx 0$. These higher degree terms have degrees 2 and 3 and can be selected as follows

> $variables := \delta\varphi(x, y, z, t), conjugate(\delta\varphi(x, y, z, t))$

$$variables := \delta\varphi, \overline{\delta\varphi} \tag{2.1.12}$$

> $0 = select(u \rightarrow degree(u, \{variables\}) :: identical(2, 3), rhs(\textbf{(2.1.10)}))$

$$0 = 2\, \varphi\, G\, \delta\varphi\, \overline{\delta\varphi} + G\, \delta\varphi^2\, \overline{\varphi} + G\, \delta\varphi^2\, \overline{\delta\varphi} \tag{2.1.13}$$

So we can either subtract these terms from **(2.1.10)**, as in **(2.1.10)** - **(2.1.13)** or, simpler, use *remove* instead of *select*, as in

> $map2(remove, u \rightarrow degree(u, \{variables\}) :: identical(2, 3), \textbf{(2.1.10)})$

$$\mathrm{i}\, \hbar\, \delta\varphi_t + \mu\, \delta\varphi = \varphi^2\, G\, \overline{\delta\varphi} + 2\, \varphi\, G\, \delta\varphi\, \overline{\varphi} + V\, \delta\varphi - \frac{\hbar^2\, \nabla^2 \delta\varphi}{2\, m} \tag{2.1.14}$$

Isolating now the term involving the time derivative and converting to *abs* only the terms involving φ we arrive at Bogoliubov's equation for $\delta\varphi$

> $isolate(\textbf{(2.1.14)}, \mathrm{i}\, \hbar\, diff(\delta\varphi(x, y, z, t), t)) :$

> $convert(\%, \mathrm{abs}, \{\varphi\}) :$

> $collect(\%, \delta\varphi)$

$$\mathrm{i}\, \hbar\, \delta\varphi_t = \left(2\, G\, |\varphi|^2 + V - \mu\right) \delta\varphi + \varphi^2\, G\, \overline{\delta\varphi} - \frac{\hbar^2\, \nabla^2 \delta\varphi}{2\, m} \tag{2.1.15}$$

The equation for $\overline{\delta\varphi}$ can now be obtained by taking the conjugate of the equation for $\delta\varphi$, expanding and collecting terms

> $-conjugate(\textbf{(2.1.15)}) :$

> $collect(expand(\%), conjugate)$

$$\mathrm{i}\, \hbar\, \overline{\delta\varphi}_t = \left(-2\, G\, |\varphi|^2 - V + \mu\right) \overline{\delta\varphi} - G\, \delta\varphi\, \overline{\varphi}^2 + \frac{\hbar^2\, \nabla^2 \overline{\delta\varphi}}{2\, m} \tag{2.1.16}$$

---

b) Solving Bogoliubov's equations **(2.1.15)** and **(2.1.16)** is, in general, a difficult problem. The computation is here restricted to a perturbation $\delta\varphi$ around the uniform solution **(1.1.5)** derived in the previous section. So we take $V = 0$, $\mu = G\, n$, and $\varphi(x, y, z) = \sqrt{n}$, where n is a constant uniform particle density. The dispersion relations for **(2.1.15)** and **(2.1.16)** are then derived using two different approaches: first simplify the problem to one in 1+1 spacetime dimensions and use Fourier transforms, then consider the whole problem in 3+1 spacetime dimensions and find the dispersion relations treating $\delta\varphi$ and $\overline{\delta\varphi}$ as independent perturbation functions.

Introduce $V = 0$, $\mu = G\, n$, and $\varphi(x, y, z) = \sqrt{n}$ into Bogoliubov's equations

> $conditions := [V(x, y, z, t) = 0, \varphi(x, y, z) = \sqrt{n}, \mu = G\, n]$

(2.1.17)

$$conditions := \left[V=0, \varphi=\sqrt{n}, \mu = G\, n\right] \quad \mathbf{(2.1.17)}$$

> *eval*(**(2.1.15)**, *conditions*) assuming $n \geq 0$

$$\mathrm{i}\, \hbar\, \delta\varphi_t = G\, n\, \delta\varphi + n\, G\, \overline{\delta\varphi} - \frac{\hbar^2\, \nabla^2 \delta\varphi}{2\, m} \quad \mathbf{(2.1.18)}$$

> *eval*(**(2.1.16)**, *conditions*) assuming $n \geq 0$

$$\mathrm{i}\, \hbar\, \overline{\delta\varphi}_t = -n\, G\, \overline{\delta\varphi} - G\, n\, \delta\varphi + \frac{\hbar^2\, \nabla^2 \overline{\delta\varphi}}{2\, m} \quad \mathbf{(2.1.19)}$$

At this point we need to introduce an expression for the perturbation $\delta\varphi$. We start considering this perturbation as a complex function, that is $\delta\varphi = u(x,t) + \mathrm{i}\, v(x,t)$, where u and v are real functions of x and t.

> *PDEtools*:-*declare*($(u, v)(x, t)$)

$u(x, t)$ *will now be displayed as* u

$v(x, t)$ *will now be displayed as* v **(2.1.20)**

> *Setup*(*realobjects* = { $(u, v)(x, t)$ })

$$\left[realobjects = \{\hbar, G, \widehat{i}, \widehat{j}, \widehat{k}, \widehat{\phi}, \widehat{r}, \widehat{\rho}, \widehat{\theta}, k, m, \mu, n, \omega, \phi, r, \rho, t, \theta, x, y, z, V, u, v\}\right] \quad \mathbf{(2.1.21)}$$

So the simplified equation **(2.1.18)** becomes

> *eval*(**(2.1.18)**, $\delta\varphi(x, y, z, t) = u(x, t) + \mathrm{i}\, v(x, t)$)

$$\mathrm{i}\, \hbar\, \left(u_t + \mathrm{i}\, v_t\right) = G\, n\, (u + \mathrm{i}\, v) + n\, G\, (u - \mathrm{i}\, v) - \frac{\hbar^2\, \nabla^2 (u + \mathrm{i}\, v)}{2\, m} \quad \mathbf{(2.1.22)}$$

Compute now the Laplacian on the right-hand side

> *expand*(*value*(**(2.1.22)**))

$$\mathrm{i}\, \hbar\, u_t - \hbar\, v_t = 2\, G\, u\, n - \frac{\hbar^2\, u_{x,x}}{2\, m} - \frac{\mathrm{i}\, \hbar^2\, v_{x,x}}{2\, m} \quad \mathbf{(2.1.23)}$$

All the objects in this equation are real, so **(2.1.23)** represents two equations, respectively conformed by its real and imaginary parts

> *expand*(Re(**(2.1.23)**))

$$-\hbar\, v_t = 2\, G\, u\, n - \frac{\hbar^2\, u_{x,x}}{2\, m} \quad \mathbf{(2.1.24)}$$

> *expand*(Im(**(2.1.23)**))

$$\hbar\, u_t = -\frac{\hbar^2\, v_{x,x}}{2\, m} \quad \mathbf{(2.1.25)}$$

To obtain the dispersion relations, take the Fourier transforms of these two equations towards eliminating $u(x, t)$ or $v(x, t)$. The Fourier transform with respect to x gives:

> *with*(*inttrans*) :

> *fourier*(**(2.1.24)**, x, k)

$$-\hbar\, \left(\mathcal{F}_{x,k}(v)\right)_t = \frac{\left(\hbar^2\, k^2 + 4\, G\, n\, m\right) \mathcal{F}_{x,k}(u)}{2\, m} \quad \mathbf{(2.1.26)}$$

> *fourier*(**(2.1.25)**, x, k)

(2.1.27)

$$\hbar \left(\mathcal{F}_{x,k}(u)\right)_t = \frac{\hbar^2 k^2 \mathcal{F}_{x,k}(v)}{2\,m} \quad \textbf{(2.1.27)}$$

Equation **(2.1.27)** can be simplified using **(2.1.26)** to eliminate $\mathcal{F}_{x,k}(u) = \mathit{fourier}(u(x,t), x, k)$ in favor of $\mathcal{F}_{x,k}(v)$ (see simplify with respect to side relations)

> *simplify*(**(2.1.27)** , {**(2.1.26)**}, {*fourier*(*u*(*x*, *t*), *x*, *k*)})

$$-\frac{2\,\hbar^2 \left(\mathcal{F}_{x,k}(v)\right)_{t,t} m}{\hbar^2 k^2 + 4\,G\,n\,m} = \frac{\hbar^2 k^2 \mathcal{F}_{x,k}(v)}{2\,m} \quad \textbf{(2.1.28)}$$

The equation above now involves only $\mathcal{F}_{x,k}(v)$. This dependency can also be eliminated by taking another fourier transform of both sides of **(2.1.28)**, now with respect to *t*, introducing the frequency ω_k

> *fourier*(**(2.1.28)**, *t*, omega[*k*])

$$\frac{2\,\hbar^2 m\,\omega_k^2 \mathcal{F}_{t,\omega_k}\left(\mathcal{F}_{x,k}(v)\right)}{\hbar^2 k^2 + 4\,G\,n\,m} = \frac{\hbar^2 k^2 \mathcal{F}_{t,\omega_k}\left(\mathcal{F}_{x,k}(v)\right)}{2\,m} \quad \textbf{(2.1.29)}$$

We see now that the double Fourier transform is a common factor of the left and right-hand sides. The desired dispersion relation for ω_k is then obtained just isolating ω_k^2

> *isolate*(**(2.1.29)**, omega[*k*]^2)

$$\omega_k^2 = \frac{k^2 \left(\hbar^2 k^2 + 4\,G\,n\,m\right)}{4\,m^2} \quad \textbf{(2.1.30)}$$

Alternatively, a 3+1 spacetime treatment is also possible treating $\delta\varphi$ and $\overline{\delta\varphi}$ as independent perturbation functions, introducing them as a plane waves $\delta\varphi = u(t)\,e^{i\,\vec{k}\cdot\vec{r}}$. To reuse *u* and *v* this time as non-real functions of *t*, unset $u(x,t)$ and $v(x,t)$ used as real objects in the previous approach

> *Setup*(*realobjects* = {*unset*, *u*(*x*, *t*), *v*(*x*, *t*)})

$$\left[\mathit{realobjects} = \{\hbar, G, \hat{i}, \hat{j}, \hat{k}, \hat{\phi}, \hat{r}, \hat{\rho}, \hat{\theta}, k, m, \mu, n, \omega, \phi, r, \rho, t, \theta, x, y, z, V\}\right] \quad \textbf{(2.1.31)}$$

> *PDEtools*:-*declare*((*u*, *v*)(*t*))

u(*t*) *will now be displayed as u*

v(*t*) *will now be displayed as v* **(2.1.32)**

We use $u(t)$ for $\delta\varphi$ and $v(t)$ for $\overline{\delta\varphi}$

> $\delta\varphi(x, y, z, t) = u(t)\exp(i\,k\_ . r\_)$;

$$\delta\varphi = u\,e^{i\left(\vec{k}\cdot\vec{r}\right)} \quad \textbf{(2.1.33)}$$

> *conjugate*($\delta\varphi(x, y, z, t)$) = $v(t)\exp(i\,k\_ . r\_)$

$$\overline{\delta\varphi} = v\,e^{i\left(\vec{k}\cdot\vec{r}\right)} \quad \textbf{(2.1.34)}$$

> k_ = k[x]_i + k[y]_j + k[z]_k

$$\vec{k} = k_x\,\hat{i} + k_y\,\hat{j} + k_z\,\hat{k} \quad \textbf{(2.1.35)}$$

> r_ = x_i + y_j + z_k

$$\vec{r} = x\,\hat{i} + y\,\hat{j} + z\,\hat{k} \tag{2.1.36}$$

where $\vec{k}$ and $\vec{r}$ are real

> Setup(realobjects = {n, k_, r_, k[x], k[y], k[z]})

$$\left[realobjects = \left\{\hbar, G, \hat{i}, \hat{j}, \hat{k}, \hat{\phi}, \hat{r}, \hat{\rho}, \hat{\theta}, k, \vec{k}, m, \mu, n, \omega, \phi, r, \vec{r}, \rho, t, \theta, x, y, z, k_x, k_y, k_z, V\right\}\right] \tag{2.1.37}$$

Introducing now the form for the perturbations **(2.1.32)** and **(2.1.33)**, expanding the scalar product and evaluating the Laplacian on the right-hand side, we have:

> eval(**(2.1.18)**, [**(2.1.33)**, **(2.1.34)**])

$$\mathrm{i}\,\hbar\,u_t\,\mathrm{e}^{\mathrm{i}\,(\vec{k}\cdot\vec{r})} = G\,n\,u\,\mathrm{e}^{\mathrm{i}\,(\vec{k}\cdot\vec{r})} + n\,G\,v\,\mathrm{e}^{\mathrm{i}\,(\vec{k}\cdot\vec{r})} - \frac{\hbar^2\,\nabla^2\left(u\,\mathrm{e}^{\mathrm{i}\,(\vec{k}\cdot\vec{r})}\right)}{2\,m} \tag{2.1.38}$$

> eval(**(2.1.38)**, [**(2.1.35)**, **(2.1.36)**])

$$\mathrm{i}\,\hbar\,u_t\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} = G\,n\,u\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} + n\,G\,v\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} - \frac{\hbar^2\,\nabla^2\left(u\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)}\right)}{2\,m} \tag{2.1.39}$$

> value(%)

$$\mathrm{i}\,\hbar\,u_t\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} = G\,n\,u\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} + n\,G\,v\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)} + \frac{\hbar^2\,u\,\mathrm{e}^{\mathrm{i}\,(x\,k_x + y\,k_y + z\,k_z)}\left(k_x^2 + k_y^2 + k_z^2\right)}{2\,m} \tag{2.1.40}$$

In the equation above, the $\mathrm{e}^{\vec{k}\cdot\vec{r}}$ is now a common factor. Isolating the time derivative of $u(t)$

> normal(isolate(**(2.1.40)**, diff(u(t), t)))

$$u_t = \frac{-\frac{\mathrm{i}}{2}\left(\hbar^2\,u\,k_x^2 + \hbar^2\,u\,k_y^2 + \hbar^2\,u\,k_z^2 + 2\,n\,G\,v\,m + 2\,G\,n\,u\,m\right)}{m\,\hbar} \tag{2.1.41}$$

Introducing $k_x^2 + k_y^2 + k_z^2 = \|\vec{k}\|^2$,

> map(u → Norm(u)^2, **(2.1.35)**)

$$\|\vec{k}\|^2 = k_x^2 + k_y^2 + k_z^2 \tag{2.1.42}$$

> algsubs((rhs = lhs)(**(2.1.42)**), **(2.1.41)**)

$$u_t = \frac{-\frac{\mathrm{i}}{2}\left(\hbar^2\,u\,\|\vec{k}\|^2 + 2\,n\,G\,v\,m + 2\,G\,n\,u\,m\right)}{m\,\hbar} \tag{2.1.43}$$

> collect(i ħ **(2.1.43)**, [G, n, m])

$$\mathrm{i}\,\hbar\,u_t = (v + u)\,n\,G + \frac{\hbar^2\,u\,\|\vec{k}\|^2}{2\,m} \tag{2.1.44}$$

An equation for the time derivative of v(t) can be obtained, for instance, by taking the conjugate of **(2.1.44)** and noting that, from **(2.1.33)** and **(2.1.34)**, $\bar{u} = v\left(\left(e^{i(\vec{k}\cdot\vec{r})}\right)\right)^2$ and $\bar{v} = u\left(\left(e^{i(\vec{k}\cdot\vec{r})}\right)\right)^2$, or, for simplicity, just repeating the steps after **(2.1.36)**, this time departing from the equation for $\delta\varphi$ **(2.1.19)**

> *eval*(**(2.1.19)**, [**(2.1.33)**, **(2.1.34)**])

$$i\hbar v_t e^{i(\vec{k}\cdot\vec{r})} = -n G v e^{i(\vec{k}\cdot\vec{r})} - G n u e^{i(\vec{k}\cdot\vec{r})} + \frac{\hbar^2 \nabla^2\left(v e^{i(\vec{k}\cdot\vec{r})}\right)}{2m} \quad (2.1.45)$$

> *value*(*eval*(**(2.1.45)**, [**(2.1.35)**, **(2.1.36)**])) :
> *normal*(*isolate*(%, *diff*(v(t), t)))

$$v_t = \frac{\frac{i}{2}\left(\hbar^2 v k_x^2 + \hbar^2 v k_y^2 + \hbar^2 v k_z^2 + 2 n G v m + 2 G n u m\right)}{m\hbar} \quad (2.1.46)$$

> *algsubs*((*rhs* = *lhs*)(**(2.1.42)**), **(2.1.46)**) :
> *collect*(i ħ %, [G, n, m])

$$i\hbar v_t = (-v - u) n G - \frac{\hbar^2 \left\|\vec{k}\right\|^2 v}{2m} \quad (2.1.47)$$

Taken together, the 1st order linear differential equations for $u(t)$ and $v(t)$, equations **(2.1.44)** and **(2.1.47)** can be rewritten as a matrix equation can be rewritten: $i\hbar \begin{bmatrix} \dot{u}(t) \\ \dot{v}(t) \end{bmatrix} = M \begin{bmatrix} u(t) \\ v(t) \end{bmatrix}$ where

$$> M := \begin{bmatrix} \frac{\hbar^2 k^2}{2m} + G n & G n \\ -G n & -\left(G n + \frac{\hbar^2 k^2}{2m}\right) \end{bmatrix} :$$

and we are using $\left\|\vec{k}\right\|^2 = k^2$. This DE system can be solved by passing it to dsolve, and from there conclude about the dispersion relations by inspection, or one can directly get the dispersion relations noting that, by definition, $\hbar\,\omega_k = \varepsilon_k$, where ε_k is the energy, in turn equal to the eigenvalues of M, so

> ε_k = *LinearAlgebra:-Eigenvalues*(M)

$$\varepsilon_k = \begin{bmatrix} \frac{\sqrt{\hbar^2 k^2 + 4 G n m}\, k\hbar}{2m} \\ -\frac{\sqrt{\hbar^2 k^2 + 4 G n m}\, k\hbar}{2m} \end{bmatrix} \quad (2.1.48)$$

Taking into account that $\varepsilon_k = \hbar\,\omega_k$, this result is the same one obtained using Fourier transforms **(2.1.30)**.

References

[1] Gross-Pitaevskii equation (wiki)
[2] Continuity equation (wiki)
[3] Bose–Einstein condensate (wiki)
[4] Dispersion relations (wiki)
[5] Advances In Atomic Physics: An Overview, Claude Cohen-Tannoudji and David Guery-Odelin, World Scientific (2011), ISBN-10: 9812774963.
[6] Nonlinear Fiber Optics, Fifth Edition (Optics and Photonics), Govind Agrawal, Academic Press (2012), ISBN-13: 978-0123970237.